

-| Corrigé du TD 5 : Séries Entières, Séries de Fourier et Intégration

| Jean Sébastien ROY, 1999

-| Ex 1 : Minimisation d'une intégrale

```
[ > restart:
[ Le degré du polynôme :
[ > N:=4;
[                                     N := 4
[ Un polynôme unitaire de degré N :
[ > P:=add(a.i*X^i,i=0..N-1)+X^N;
[                                     P := a0 + a1 X + a2 X^2 + a3 X^3 + X^4
[ On intègre (pas la peine de prendre la racine carrée pour minimiser) :
[ > E:=int(P*P,X=0..1);
[                                     E := 1/9 + 1/4 a3 + 2/7 a2 + 1/7 a3^2 + 1/3 a1 + 1/3 a2 a3 + 2/5 a0 + 2/5 a1 a3 + 1/5 a2^2 + 1/2 a0 a3
[                                     + 1/2 a1 a2 + 2/3 a0 a2 + 1/3 a1^2 + a0 a1 + a0^2
[ On résoud le système formé de l'ensemble des dérivées partielles de E :
[ (Si il existe un polynôme qui minimise E, alors ce polynôme vérifie ce système)
[ > S:={seq(diff(E,a.i),i=0..3)};
[ S := { 2/5 + 1/2 a3 + 2/3 a2 + a1 + 2 a0, 1/3 + 2/5 a3 + 1/2 a2 + 2/3 a1 + a0,
[       2/7 + 1/3 a3 + 2/5 a2 + 1/2 a1 + 2/3 a0, 1/4 + 2/7 a3 + 1/3 a2 + 2/5 a1 + 1/2 a0 }
[ > sol:=solve(S);
[                                     sol := { a1 = -2/7, a2 = 9/7, a3 = -2, a0 = 1/70 }
[ Il n'y a qu'une solution.
[ Calculons la hessienne au point pour déterminer si c'est bien un minimum local :
[ > H:=matrix(N,N,(i,j)->diff(E,a.(i-1),a.(j-1)));
[                                     H := [ 2      1      2/3     1/2 ]
[                                     [ 1      2/3     1/2      2/5 ]
[                                     [ 2/3     1/2      2/5      1/3 ]
[                                     [ 1/2     2/5      1/3      2/7 ]
[ Assez logiquement, H ne dépend pas du point. Est-elle définie positive ?
[ > with(linalg):
[ Warning, new definition for norm
[ Warning, new definition for trace
[ > definite(H,'positive_def');
[                                     true
[ On peut vérifier le résultat ainsi :
[ > evalf(allvalues(eigenvalues(H)));
[ 3.000428560 + .1118956086 10^-10 I, .338282440 - .1600754195 10^-11 I,
[ .01347652259 + .8809869351 10^-9 I, .000193429411 - .8905757417 10^-9 I
[ Donc le polynôme minimisant l'intégrale :
[ > subs(sol,P);
```

$$\frac{1}{70} - \frac{2}{7}X + \frac{9}{7}X^2 - 2X^3 + X^4$$

[Et la valeur cherchée :

[> `sqrt(subs(sol,E));`

$$\frac{1}{210}$$

[On peut aussi utiliser la fonction minimize à titre de vérification :

[> `minimize(sqrt(E));`

$$\frac{1}{210}$$

- Ex 2 : Calcul d'intégrale par intégration par parties et DSE

[> `restart;`

[> `f:=t->ln(t)*ln(1-t)/t;`

$$f := t \rightarrow \frac{\ln(t) \ln(1-t)}{t}$$

[> `A:=int(f(t),t=0..1);`

$$A := \zeta(3)$$

[Bon. Maple y arrive tout seul (parfois). Pour le forcer à ne pas faire le calcul on utilise Int au lieu de int :

[> `A:=Int(f(t),t=0..1);`

$$A := \int_0^1 \frac{\ln(t) \ln(1-t)}{t} dt$$

[Vérifions l'existence de l'intégrale en 0 :

[> `series(f(t),t=0,2);`

$$-\ln(t) + O(t)$$

[Ok. Intégration par parties :

[> `student[intparts](A,ln(t)*ln(1-t));`

$$-\int_0^1 \left(\frac{\ln(1-t)}{t} - \frac{\ln(t)}{1-t} \right) \ln(t) dt$$

[> `expand(%);`

$$-\int_0^1 \frac{\ln(t) \ln(1-t)}{t} dt + \int_0^1 \frac{\ln(t)^2}{1-t} dt$$

[Il reste donc à calculer l'intégrale à droite (qui vaut 2*A), ce que l'on fait via un DSE de 1/(1-t) :

[> `S:=n->int(t^n*(ln(t))^2/2,t=0..1);`

$$S := n \rightarrow \int_0^1 \frac{1}{2} t^n \ln(t)^2 dt$$

[> `S(n);`

$$\lim_{t \rightarrow 0^+} -\frac{1}{2} (t \ln(t)^2 e^{(n \ln(t))} n^2 - 2 t \ln(t) e^{(n \ln(t))} n + 2 t \ln(t)^2 e^{(n \ln(t))} n - 2 t \ln(t) e^{(n \ln(t))} - 2 + t \ln(t)^2 e^{(n \ln(t))} + 2 t e^{(n \ln(t))}) / ((n+1)(1+2n+n^2))$$

[Stupide, Maple n'arrive pas à simplifier cette expression (meme avec des assume adéquats), donc on l'aide :

[> `op(1,%);`

$$-\frac{1}{2} (t \ln(t)^2 e^{(n \ln(t))} n^2 - 2 t \ln(t) e^{(n \ln(t))} n + 2 t \ln(t)^2 e^{(n \ln(t))} n - 2 t \ln(t) e^{(n \ln(t))} - 2 + t \ln(t)^2 e^{(n \ln(t))} + 2 t e^{(n \ln(t))}) / ((n+1)(1+2n+n^2))$$

[Manifestement, seul le terme '-2' du numérateur ne tends pas vers zéro, donc :

[> **subs(t=0,%);**

$$\frac{1}{(n+1)(1+2n+n^2)}$$

[> **factor(%);**

$$\frac{1}{(n+1)^3}$$

[D'où la valeur de l'intégrale :

[> **S:=Sum(1/n^3,n=1..infinity):A=S;**

$$\int_0^1 \frac{\ln(t) \ln(1-t)}{t} dt = \sum_{n=1}^{\infty} \frac{1}{n^3}$$

[> **S=value(S);**

$$\sum_{n=1}^{\infty} \frac{1}{n^3} = \zeta(3)$$

- Ex 3 : DSE utilisant une équation différentielle

[> **restart:with(powseries):**

[> **f:=x->(arcsin(x/2))^2;**

$$f := x \rightarrow \arcsin\left(\frac{1}{2}x\right)^2$$

[Le package Powerseries ne traite pas la fonction arcsin. On va dériver arcsin(x/2), calculer le DSE puis intégrer le résultat.

[> **g:=unapply(diff(arcsin(x/2),x)):g(x);**

$$\frac{1}{\sqrt{4-x^2}}$$

[On calcule le DSE de g, on l'intègre, et on l'élève au carré :

[> **evalpow(g(x)):powint(%):S:=evalpow(%^2):**

[> **tpsform(%,x,10);**

$$\frac{1}{4}x^2 + \frac{1}{48}x^4 + \frac{1}{360}x^6 + \frac{1}{2240}x^8 + O(x^{10})$$

[Vérifions à partir d'un DL de f :

[> **series(f(x),x=0,9);**

$$\frac{1}{4}x^2 + \frac{1}{48}x^4 + \frac{1}{360}x^6 + \frac{1}{2240}x^8 + O(x^{10})$$

[Déterminons une équation différentielle. à coefficients polynômiaux dont f est solution :

[> **eq1:=dy=diff(f(x),x):eq1;**

$$dy = 2 \frac{\arcsin\left(\frac{1}{2}x\right)}{\sqrt{4-x^2}}$$

[> **eq2:=dy2=diff(f(x),x\$2):eq2;**

$$dy2 = 2 \frac{1}{4-x^2} + 2 \frac{\arcsin\left(\frac{1}{2}x\right)x}{(4-x^2)^{(3/2)}}$$

[On isole $\arcsin(x/2)$ dans la première équation...

[> `student[isolate](eq1,arcsin(1/2*x));`

$$\arcsin\left(\frac{1}{2}x\right) = \frac{1}{2} dy \sqrt{4-x^2}$$

[Et l'on substitue dans la seconde :

[> `subs(%,eq2);`

$$dy2 = 2 \frac{1}{4-x^2} + \frac{dy x}{4-x^2}$$

[La voilà. $dy2$ est f' et dy est f .

[> `eqy:=collect(numer(rhs(%)-lhs(%)),dy2):eqy=0;`

$$(4-x^2) dy2 - 2 - dy x = 0$$

[On cherche alors une solution DSE :

[> `dy:=sum(a(k)*x^k,k=0..N);`

$$dy := \sum_{k=0}^N a(k) x^k$$

[> `dy2:=diff(dy,x);`

[> `ex:=simplify(combine(eqy));`

$$ex := \left(\sum_{k=0}^N (4 a(k) k x^{k-1} - a(k) k x^{k+1} - a(k) x^{k+1}) \right) - 2$$

[Les conditions initiales :

[> `collect(expand(subs(N=3,ex)),x);`

$$-4 a(3) x^4 - 3 a(2) x^3 + (-2 a(1) + 12 a(3)) x^2 + (-a(0) + 8 a(2)) x + 4 a(1) - 2$$

[Donc $a(1) = 1/2$, ...

[A partir de cette expression, on cherche à obtenir une relation de récurrence vérifiée par les coefficients du DSE.

[Il faut un peu manipuler à la main :

[> `rel[A]:=op([1,1],ex);`

$$rel_A := 4 a(k) k x^{k-1} - a(k) k x^{k+1} - a(k) x^{k+1}$$

[On s'arrange pour avoir les puissances de x toutes égales :

[> `subs(k=k+2,op(1,rel[A]))+op(2,rel[A])+op(3,rel[A]);`

$$4 a(k+2) (k+2) x^{k+1} - a(k) k x^{k+1} - a(k) x^{k+1}$$

[> `rel[B]:=simplify(%/x^(k+1));`

$$rel_B := 4 a(k+2) k + 8 a(k+2) - a(k) k - a(k)$$

[On exprime la relation de récurrence :

[> `rel[C]:=student[isolate](rel[B],a(k));`

$$rel_C := a(k) = \frac{-4 a(k+2) k - 8 a(k+2)}{-k-1}$$

[Après simplification et en remarquant que seuls les coefficients impairs peuvent être non nuls (par parité de f (donc f' est impaire)), on change $a(k)$ en $b(p)$ avec $k=2*p-1$:

[> `a:=k->b((k+1)/2);`

$$a := k \rightarrow b\left(\frac{1}{2}k + \frac{1}{2}\right)$$

[> `rel[D]:=eval(subs(k=2*p+1,rel[C]));`

$$rel_D := b(p+1) = \frac{-4 b(p+2) (2p+1) - 8 b(p+2)}{-2p-2}$$

[On résoud :

[> `rb:=unapply(rsolve({rel[D],b(1)=1/2},b(p)),p);`

$$rb := p \rightarrow \frac{4^{(-p)} \Gamma(p) \sqrt{\pi}}{\Gamma\left(p + \frac{1}{2}\right)}$$

[Donc pour k impair le terme du developpement de f' est :

[> `subs(p=(k+1)/2,rb(p))*x^k;`

$$\frac{4^{(-1/2 k - 1/2)} \Gamma\left(\frac{1}{2} k + \frac{1}{2}\right) \sqrt{\pi} x^k}{\Gamma\left(\frac{1}{2} k + 1\right)}$$

[On intègre :

[> `c:=unapply(subs(k=n-1,int(%,x)),n);`

$$c := n \rightarrow \frac{4^{(-1/2 n)} \Gamma\left(\frac{1}{2} n\right) \sqrt{\pi} x^n}{\Gamma\left(\frac{1}{2} n + \frac{1}{2}\right) n}$$

[Voilà le terme général du developpement :

[> `c(2*n);`

$$\frac{1}{2} \frac{4^{(-n)} \Gamma(n) \sqrt{\pi} x^{(2n)}}{\Gamma\left(n + \frac{1}{2}\right) n}$$

[Pas de chance, Maple ne sait pas simplifier plus. Comparons avec la solution connue (notée v dans la suite) :

[> `v:=n->1/2*(n-1)!^2/(2*n)!;`

$$v := n \rightarrow \frac{1}{2} \frac{((n-1)!)^2}{(2n)!}$$

[C'est bon :

[> `simplify(v(n)*x^(2*n)-c(2*n));`

0

[> `sum(c(2*n),n=1..4);`

$$\frac{1}{4} x^2 + \frac{1}{48} x^4 + \frac{1}{360} x^6 + \frac{1}{2240} x^8$$

[> `series(f(x),x=0,9);`

$$\frac{1}{4} x^2 + \frac{1}{48} x^4 + \frac{1}{360} x^6 + \frac{1}{2240} x^8 + O(x^{10})$$

- Ex 4 : Série de Fourier des polygones

[> `restart:with(plots):`

[> `n:=4:`

[On définit un polygone par ses sommets :

[> `z:=[1,1+I,-1+I,-I,1];`

$$z := [1, 1 + I, -1 + I, -I, 1]$$


[Puis en formant une fonction affine par morceaux, continue, qui représente le dit polygone sur $0..1$:

[> `g:=t->piecewise(seq(op([t<=j/n,n*(z[j]*(j/n-t)+z[j+1]*(t-(j-1)/n)]),j=1..n)):`

[> `g(t);`

$$\left\{ \begin{array}{ll} 1 - 4t + (4 + 4I)t & t \leq \frac{1}{4} \\ (4 + 4I) \left(\frac{1}{2} - t\right) + (-4 + 4I) \left(t - \frac{1}{4}\right) & t \leq \frac{1}{2} \\ (-4 + 4I) \left(\frac{3}{4} - t\right) - 4I \left(t - \frac{1}{2}\right) & t \leq \frac{3}{4} \\ -4I(1-t) + 4t - 3 & t \leq 1 \end{array} \right.$$

```
> pf:=complexplot(g(t),t=0..1,color=blue):display(pf);
```



[Calculons le développement en série de Fourier de g. Tout d'abord les coefficients :

```
> c:=m->int(g(t)*exp(2*I*Pi*t*m),t=0..1);
```

$$c := m \rightarrow \int_0^1 g(t) e^{2I\pi t m} dt$$

[Puis la somme partielle :

```
> s:=(t,nn)->sum('c(m)*exp(2*I*Pi*t*m)',m=-nn..nn);
```

$$s := (t, nn) \rightarrow \sum_{m=-nn}^{nn} 'c(m) e^{2I\pi t m}'$$


[Puis le graphe :

```
> p:=k->complexplot(s(t,k),t=0..1,numpoints=400,color=COLOR(RGB,(6-k)/5,0,(k-1)/5));
```

```
p := k → complexplot( s(t,k), t = 0 .. 1, numpoints = 400,
color = COLOR( RGB, ( 6/5 - 1/5 k, 0, 1/5 k - 1/5 ) ) )
```

[Et on trace les 5 premières sommes partielles ainsi que le polygone :

```
> display(p(1),p(2),p(3),p(4),p(5),pf);
```



[Ca converge (uniformément même).